#$Id$
# Maintainer: Daniel Isenmann <daniel@archlinux.org>

pkgname=sg3_utils
pkgver=1.42
pkgrel=1
pkgdesc="Generic SCSI utilities"
arch=(i686 x86_64)
url="http://sg.danny.cz/sg/sg3_utils.html"
license=('GPL' 'custom:BSD')
depends=('glibc')
source=(http://sg.danny.cz/sg/p/${pkgname}-${pkgver}.tar.xz)
md5sums=('913ac2c9069d2ba44e05565a445810ab')

build() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  ./configure --prefix=/usr
  make
}

package() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  make DESTDIR="${pkgdir}" install
  install -m755 -d "${pkgdir}/usr/share/licenses/${pkgname}"
  install -m644 COPYING "${pkgdir}/usr/share/licenses/${pkgname}/"
}
